Otimização de Sistemas

Prof. Sandro Jerônimo de Almeida, PhD.

Problema do Transporte e Problema de Atribuição

Problema do Transporte

- O problema de transporte consiste em transportar itens de *m* origens ate´ *n* destinos por canais disponíveis, de forma a atender as demandas de *n* e minimizar custo total de transporte, uma vez que, cada canal tem um custo por unidade. As *m* origens também possuem capacidade de produção.

Exemplo

Capacidade de Produção

Origens

Custos

Demanda

Destinos



Cenário

Fábricas de Enlatados

- Ribeirão Preto (SP)
- Campina Grande (PB)
- Joinville (SC)

Centro de distribuição

- São Paulo (SP)
- Goiânia (GO)
- Recife (PE)
- Manaus (AM)

Fábrica	Produção*	Centro de Distribuição	Demanda*
Ribeirão Preto (SP)	15	São Paulo (SP)	20
Campina Grande (PB)	12	Goiânia (GO)	5
Joinville (SC)	8	Recife (PE)	5
Total Produção	**35**	Manaus (AM)	5
(*) Toneladas mensais		**Demanda Total**	**35**

Distância e Custo de Transporte

Origem/ Destino	Ribeirão Preto (SP)	Campina Grande (PB)	Joinville (SC)
São Paulo (SP)	315 km	2.712 km	531 km
Goiânia (GO)	608 km	2.331 km	1.408 km
Recife (PE)	2.544 km	195 km	3.226 km
Manaus (AM)	3.657 km	4.516 km	4.230 km



Na prática, o custo deve ser por unidade transportada na rota. Exemplo:
Assumindo R$ 0,50 o custo por tonelada (Ton)
Se viajar 500 km levando 1.000kg, temos custo na rota = R$ 0,50 x 500 = R$ 250

Variáveis de Decisão

12 variáveis: quantidade de toneladas a serem escoadas em cada uma das rotas (3 origens x 4 destino)

Origem/ Destino	Ribeirão Preto (SP)	Campina Grande (PB)	Joinville (SC)
São Paulo (SP)			
Goiânia (GO)			
Recife (PE)			
Manaus (AM)			

Função Objetivo

1) Xij = quantidade de toneladas a serem transportadas entre a origem *i* e o destino *j*
2) Minimizar c = $(315 \times X_{11}) + (608 \times X_{12}) + (2.544 \times X_{13}) + (3.657 \times X_{14})$

$$+ (2.712 \times X_{21}) + (2.331 \times X_{22}) + (195 \times X_{23}) + (4.516 \times X_{24})$$

$$+ (531 \times X_{31}) + (1.408 \times X_{32}) + (3.226 \times X_{33}) + (4.230 \times X_{34})$$

Restrições de Produção

Fábrica	Produção*
Ribeirão Preto (SP)	15
Campina Grande (PB)	12
Joinville (SC)	8
Total Produção	**35**
(*) Toneladas mensais	

1) $X_{11} + X_{12} + X_{13} + X_{14} = 15$
2) $X_{21} + X_{22} + X_{23} + X_{24} = 12$
3) $X_{31} + X_{32} + X_{33} + X_{34} = 8$

Restrições de Demanda

Centro de Distribuição	Demanda*
São Paulo (SP)	20
Goiânia (GO)	5
Recife (PE)	5
Manaus (AM)	5
Demanda Total	**35**

1) $X_{11} + X_{21} + X_{31} = 20$
2) $X_{12} + X_{22} + X_{32} = 5$
3) $X_{13} + X_{23} + X_{33} = 5$
4) $X_{14} + X_{24} + X_{34} = 5$

Modelo de Otimização

1) Xij = quantidade de toneladas a serem transportadas entre a origem *i* e o destino *j*

2) Minimizar c = $(315 \times X_{11}) + (608 \times X_{12}) + (2.544 \times X_{13}) + (3.657 \times X_{14})$

$+ (2.712 \times X_{21}) + (2.331 \times X_{22}) + (195 \times X_{23}) + (4.516 \times X_{24})$

$+ (531 \times X_{31}) + (1.408 \times X_{32}) + (3.226 \times X_{33}) + (4.230 \times X_{34})$

Sujeito a:

$X_{11} + X_{12} + X_{13} + X_{14} = 15$

$X_{21} + X_{22} + X_{23} + X_{24} = 12$

$X_{31} + X_{32} + X_{33} + X_{34} = 8$

$X_{11} + X_{21} + X_{31} =20$

$X_{12} + X_{22} + X_{32} = 5$

$X_{13} + X_{23} + X_{33} =5$

$X_{14} + X_{24} + X_{34} = 5$

Variáveis devem assumir valores maiores ou igual a zero

Problema de Transporte

- Formulação Geral

$$
\begin{aligned}
Minimizar\ z &= \sum_{i=1}^{m}\sum_{j=1}^{n} c_{ij}x_{ij} \\
\sum_{j=1}^{n} x_{ij} &= o_i \qquad i = 1, \ldots, m \\
\sum_{i=1}^{m} x_{ij} &= d_j \qquad j = 1, \ldots, n \\
x_{ij} \geq 0 \quad i &= 1, \ldots, m;\ j = 1, \ldots, n
\end{aligned}
$$

Problema de Transporte - AMPL

Dados

```
# Define o conjunto "ORIG" e "oferta"
param: ORIG: oferta :=
RP 15
PB 12
SC 8 ;
# define "DEST" e "demanda"
param: DEST: demanda :=
SP 20
GO 5
PE 5
AM 5 ;

param Custo:
SP GO PE AM :=
RP 315 608 2544 3657
PB 2712 2331 195 4516
SC 531 1408 3226 4230 ;
```

Transporte_Brasil.dat

Modelo

```
set ORIG; # Origens
set DEST; # Destinos

param oferta {ORIG} >= 0; # quantidade disponível nas origens
param demanda {DEST} >= 0; # quantidade requerida nos destinos

check: sum {i in ORIG} oferta[i] = sum {j in DEST} demanda[j]; #tratamento entradas

param Custo {ORIG,DEST} >= 0; # custo de entrega (por unidade) na rota (i,j)

var X {ORIG,DEST} >= 0; # Varíaveis de decisão - unidades a serem entregues na rota (i,j)

minimize Custo_Total:
sum {i in ORIG, j in DEST} Custo[i,j] * X[i,j];
subject to Oferta {i in ORIG}:
sum {j in DEST} X[i,j] = oferta[i];
subject to Demanda {j in DEST}:
sum {i in ORIG} X[i,j] = demanda[j];
```

Problema_transporte.mod

Problema de Transporte - AMPL

- Solução

Fábrica	Produção*	Centro de Distribuição	Demanda*
Ribeirão Preto (SP)	15	São Paulo (SP)	20
Campina Grande (PB)	12	Goiânia (GO)	5
Joinville (SC)	8	Recife (PE)	5
Total Produção	**35**	Manaus (AM)	5
(*) Toneladas mensais		**Demanda Total**	**35**

Comandos

1) ampl: model problema_transporte.mod
2) ampl: data Transporte_Brasil.dat
3) ampl: solve;

MINOS 5.51: optimal solution found.
3 iterations, objective 38069

4) ampl: display X;

```
X :=
PB AM    5
PB GO    2
PB PE    5
PB SP    0
RP AM    0
RP GO    3
RP PE    0
RP SP   12
SC AM    0
SC GO    0
SC PE    0
SC SP    8 ;
```

Problema de Atribuição
Assignment problem

Consiste em alocar <u>*m* origens</u> para <u>*n* destinos</u> de forma a:

- Maximizar o benefício da alocação, ou;
- Minimizar os custos de alocação

- Variação do Problema do Transporte onde as ofertas e demandas são unitárias

Exemplo - Alocação de Projetos



Na prática, o custo de alocação = (R$) Valor da hora x N° Horas Efetivas + Risco

Valor da Hora: Desenvolvedor Mobile > Desenvolver Web > Estagiário

Variáveis de Decisão

9 variáveis: Decisões de alocar de cada profissional para cada um dos projetos (3 origens x 3 destino)

Profissional/ Projeto	WebSite	App	BackEnd
Desenvolvedor Mobile	X_{11} x 5.000	X_{12} x 6.000	X_{13} x 9.000
Desenvolvedor Web	X_{21} x 2.000	X_{22} x 9.000	X_{23} x 10.000
Estagiário	X_{31} x 7.000	X_{32} x 7.000	X_{33} x 11.000

Função Objetivo

1) X_{ij} = decisão sobre alocação do profissional *i* para o projeto j

0 → Não será alocado | 1 → Será alocado

2) Minimizar custos = $(5.000 \times X_{11}) + (6.000 \times X_{12}) + (9.000 \times X_{13})$

$+ (2.000 \times X_{21}) + (9.000 \times X_{22}) + (10.000 \times X_{23})$

$+ (7.000 \times X_{31}) + (7.000 \times X_{32}) + (11.000 \times X_{33})$

WebSite

Aplicativo

BackEnd

Restrições

1) Cada Profissional deve estar a um único projeto

$$X_{11} + X_{12} + X_{13} = 1$$

$$X_{21} + X_{22} + X_{23} = 1$$

$$X_{31} + X_{32} + X_{33} = 1$$

$$X_{11} + X_{21} + X_{31} = 1$$

WebSite

$$X_{12} + X_{22} + X_{32} = 1$$

Aplicativo

$$X_{13} + X_{23} + X_{33} = 1$$

BackEnd

2) Cada Projeto deve estar a um único Profissional

Modelo de Otimização

1) X_{ij} = decisão sobre alocação do profissional *i* para o projeto j

0 → Não será alocado | 1 → Será alocado

2) Minimizar custos = $(5.000 \times X_{11}) + (6.000 \times X_{12}) + (9.000 \times X_{13})$

$+ (2.000 \times X_{21}) + (9.000 \times X_{22}) + (10.000 \times X_{23})$

$+ (7.000 \times X_{31}) + (7.000 \times X_{32}) + (11.000 \times X_{33})$

Sujeito a:

$X_{11} + X_{12} + X_{13} = 1$
$X_{21} + X_{22} + X_{23} = 1$
$X_{31} + X_{32} + X_{33} = 1$

$X_{11} + X_{21} + X_{31} = 1$
$X_{12} + X_{22} + X_{32} = 1$
$X_{13} + X_{23} + X_{33} = 1$

$X_{ij} = \{0, 1\}$
$i = \{1, 2, 3\}$
$j = \{1, 2, 3\}$

Exercício – Problema de Atribuição

- Apresentar o modelo geral do problema de atribuição

Considerando o enunciado anterior faça

- Criar o arquivo de modelo no formato AMPL
- Criar o arquivo de dados no formato AMPL
- Apresentar a solução usando solver para AMPL

Problema de Atribuição

- Formulação Geral

$$
\begin{aligned}
Minimizar\ z &= \sum_{i=1}^{n}\sum_{j=1}^{n} c_{ij}x_{ij} \\
\sum_{j=1}^{n} x_{ij} &= 1 \qquad i = 1, \ldots, n \\
\sum_{i=1}^{n} x_{ij} &= 1 \qquad j = 1, \ldots, n \\
x_{ij} \geq 0 \quad i &= 1, \ldots, m;\ j = 1, \ldots, n
\end{aligned}
$$

Exercício – Resolva utilizando AMPL



<u>Apresente</u>: além da resposta final, apresente os arquivos do modelo (.mod) e dados (.dat)

Resolução com AMPL – Alternativa 1

Dados

```
set ORIG := DesMobile DesWeb Estagiario ;

set DEST := WebSite App BackEnd;

#Realiza ajuste na oferta e demanda

param oferta default 1 ;
param demanda default 1 ;

param Custo:
WebSite App BackEnd :=
DesMobile 5000 6000 9000
DesWeb 2000 9000 10000
Estagiario 7000 7000 11000 ;
```

Atribuicao_com_transporte.dat

Modelo de Transporte

```
set ORIG; # Origens
set DEST; # Destinos

param oferta {ORIG} >= 0; # quantidade disponível nas origens
param demanda {DEST} >= 0; # quantidade requerida nos destinos

check: sum {i in ORIG} oferta[i] = sum {j in DEST} demanda[j]; #tratamento entradas

param Custo {ORIG,DEST} >= 0; # custo de entrega (por unidade) na rota (i,j)

var X {ORIG,DEST} >= 0; # Varíaveis de decisão - unidades a serem entregues na rota (i,j)

minimize Custo_Total:
sum {i in ORIG, j in DEST} Custo[i,j] * X[i,j];
subject to Oferta {i in ORIG}:
sum {j in DEST} X[i,j] = oferta[i];
subject to Demanda {j in DEST}:
sum {i in ORIG} X[i,j] = demanda[j];
```

Problema_transporte.mod

Para solução inteira use o *CPLEX*
ampl: option solver cplex;

Resolução com AMPL – Alternativa 2

Dados

```
set PROFISSIONAL := DesMobile
                    DesWeb
                    Estagiario ;

set TAREFA := WebSite App BackEnd;

param Custo:
WebSite App BackEnd :=
DesMobile 5000 6000 9000
DesWeb 2000 9000 10000
Estagiario 7000 7000 11000 ;
```

Atribuicao_profissionais.dat

Modelo de Atribuição

```
set PROFISSIONAL; # Origens
set TAREFA; # Destinos

# custo de atribuição
param Custo {PROFISSIONAL,TAREFA} >= 0;

var X {PROFISSIONAL,TAREFA} binary; # Varíaveis de decisão

minimize Custo_Total:
sum {i in PROFISSIONAL, j in TAREFA} Custo[i,j] * X[i,j];

subject to Profissionais {i in PROFISSIONAL}:
sum {j in TAREFA} X[i,j] = 1;

subject to Tarefas {j in TAREFA}:
sum {i in PROFISSIONAL} X[i,j] = 1;
```

Problema_Atribuicao.mod

Para solução inteira use o *CPLEX*
ampl: option solver cplex;